К. Ю. Осипенко

СТАНДАРТНАЯ ПРОГРАММА ВЫЧИСЛЕНИЯ ИНТЕГРАЛОВ ФРЕНЕЛЯ ПРОГРАММА $SFFRCS(C, S, X)$

§1. Назначение программы

Программа служит для вычисления значений интегралов Френеля:

$$C(x) = \frac{1}{\sqrt{2\pi}} \int_0^x \frac{\cos t}{\sqrt{t}}\,dt,$$
$$S(x) = \frac{1}{\sqrt{2\pi}} \int_0^x \frac{\sin t}{\sqrt{t}}\,dt.$$

§2. Обращение

Обращение к программе осуществляется с помощью оператора $CALL$:

$$CALL\textvisiblespace SFFRCS(C, S, X)$$

C — результат, значение $C(|x|)$,
S — результат, значение $S(|x|)$,
X — аргумент x.

§3. Метод

Для вычисления интегралов Френеля используется следующий метод [1]
1. При $0 \le x \le 8$:

$$C(x) = \frac{1}{\sqrt{2\pi}} \int_0^x \frac{\cos t}{\sqrt{t}}\,dt = \sqrt{\frac{x}{2\pi}} \int_0^1 \frac{\cos ux}{\sqrt{u}}\,du,$$
$$S(x) = \frac{1}{\sqrt{2\pi}} \int_0^x \frac{\sin t}{\sqrt{t}}\,dt = \sqrt{\frac{x}{2\pi}} \int_0^1 \frac{\sin ux}{\sqrt{u}}\,du.$$

Разлагая $\cos ux$ и $\sin ux$ в ряды по многочленам Чебышева $T_n\left(\frac{x}{8}\right)$ и интегрируя их, получим:

$$C(x) \approx \sqrt{\frac{x}{2\pi}} \left[\rho_0 + 2\sum_{n=1}^{12} (-1)^n \rho_{2n} T_{2n}\left(\frac{x}{8}\right)\right],$$
$$S(x) \approx \sqrt{\frac{x}{2\pi}} \left[2\sum_{n=1}^{12} (-1)^n \rho_{2n+1} T_{2n+1}\left(\frac{x}{8}\right)\right].$$

2. При $x \geq 8$:

$$C(x) + iS(x) = \frac{1}{\sqrt{2\pi}} \int_0^x \frac{e^{it}}{\sqrt{t}}\, dt = \frac{1}{\sqrt{2\pi}} \int_0^\infty \frac{e^{it}}{\sqrt{t}}\, dt - \frac{1}{\sqrt{2\pi}} \int_x^\infty \frac{e^{it}}{\sqrt{t}}\, dt$$
$$= \frac{1}{2}(1+i) - \frac{1}{\sqrt{2\pi}} \int_x^\infty \frac{e^{it}}{\sqrt{t}}\, dt,$$

сделаем замену: $t = \dfrac{(u+x)^2}{x}$, тогда:

$$\frac{1}{\sqrt{2\pi}} \int_0^x \frac{e^{it}}{\sqrt{t}}\, dt = \frac{e^{ix}}{\sqrt{2\pi x}} \int_0^\infty 2e^{i\left(\frac{u^2}{x}+2u\right)}\, du.$$

Преобразуем интеграл в правой части этого равенства с помощью замены переменного $u = it$, затем $t = \sqrt{8}\eta$:

$$\int_0^\infty 2e^{i\left(\frac{u^2}{x}+2u\right)}\, du = i\int_0^\infty 2e^{-i\frac{t^2}{x}} e^{-2t}\, dt = i\int_0^\infty 2\sqrt{8}e^{-i\frac{8}{x}\eta^2} e^{-2\sqrt{8}\eta}\, d\eta.$$

Обозначим:

$$(*) \qquad \begin{aligned} A\left(\frac{8}{x}\right) &= \int_0^\infty 2\sqrt{8}e^{-2\sqrt{8}\eta} \cos\frac{8}{x}\eta^2\, d\eta, \\ B\left(\frac{8}{x}\right) &= \int_0^\infty 2\sqrt{8}e^{-2\sqrt{8}\eta} \sin\frac{8}{x}\eta^2\, d\eta, \end{aligned}$$

тогда

$$C(x) + iS(x) = \frac{1}{2}(1+i) - \frac{\cos x + i\sin x}{\sqrt{2\pi x}} \left(iA\left(\frac{8}{x}\right) + B\left(\frac{8}{x}\right)\right).$$

Таким образом,

$$C(x) = \frac{1}{2} + \frac{\sin x}{\sqrt{2\pi x}} A\left(\frac{8}{x}\right) - \frac{\cos x}{\sqrt{2\pi x}} B\left(\frac{8}{x}\right),$$
$$S(x) = \frac{1}{2} - \frac{\sin x}{\sqrt{2\pi x}} B\left(\frac{8}{x}\right) - \frac{\cos x}{\sqrt{2\pi x}} A\left(\frac{8}{x}\right).$$

Разлагая $\cos\dfrac{8}{x}\eta^2$ и $\sin\dfrac{8}{x}\eta^2$ в ряды по многочленам Чебышева $T_n\left(\dfrac{8}{x}\right)$ и подставляя полученные ряды в (*), имеем:

$$A\left(\frac{8}{x}\right) \approx \gamma_0 + 2\sum_{n=1}^{12} (-1)^n \gamma_{2n} T_{2n}\left(\frac{8}{x}\right),$$

$$B\left(\frac{8}{x}\right) \approx 2\sum_{n=0}^{12} (-1)^n \gamma_{2n+1} T_{2n+1}\left(\frac{8}{x}\right).$$

Значения коэффициентов ρ_i и γ_i ($i = 0, \ldots, 25$) приведены в [1]. Для вычисления линейных комбинаций многочленов Чебышева используется модифицированный алгоритм, предложенный в [2].

Алгоритм обеспечивает на машине БЭСМ-6 10 верных значащих цифр.

§4. Текст программы

```
      SUBROUTINE SFFRCS(C,S,X)
      DIMENSION A(52),RK(13),RL(13)
      REAL X,C,S,Z,H,Y,F,D,E,B,R,T
      INTEGER K,L
      DATA (A=.1E-10,-.366E-9,.10898E-7,-.267681E-6,.527608E-5,
     *-.81056841E-4,.933990129E-3,-.7651297534E-2,.041140949487,
     *4E-11,-.128E-9,.4206E-8,-.11507E-6,.2562196E-5,
     *-,45321924E-4,.617420236E-3,-.6220184292E-2,.043868192558,
     *-.200717449332,.53866661798,-.799616840492,1.053859157204,
     *.1E-11,-.4E-11,.14E-10,-.54E-10,.239E-9,-.1176E-8,.6545E-8,
     *-.42829E-7,.347441E-6,-.3810219E-5,.66275081E-4,
     *-.2617529549E-2,.994548822473,.2E-11,-.6E-11,.18E-10,
     *-.72E-10,.298E-9,-.1346E-8,.6798E-8,-.39518E-7,.275996E-6,
     *-.2475448E-5,.3202967E-4,-.755202944E-3,.06088192415))
      Z=ABS(X)
      IF (Z-8.) 1, 2,2
    1 H=Z/8.
      K=0
      GOTO 3
    2 H=8./Z
      K=26
    3 L=K+13
      Y=4.*H*H-2.
      RK(1)=A(1+K)
      RK(2)=Y*RK(1)+A(2+K)
      RL(1)=A(1+L)
      RL(2)=Y*RL(1)+A(2+L)
      DO 4 I=3,13
      RK(I)=Y*RK(I-1)-RK(I-2)+A(I+K)
      RL(I)=Y*RL(I-1)-RL(I-2)+A(I+L)
    4 CONTINUE
      F=.398942280401
      D=F*RK(13)
      E=F*RL(13)*H
      B=SQRT(Z)
      IF (Z-8.) 5, 6,6
    5 C=D*B
      S=E*B
      GOTO 7
    6 R=SIN(Z)
```

```
      T = COS(Z)
      C = 0.5 + (D * R - E * T)/B
      S = 0.5 - (E * R + D * T)/B
    7 RETURN
      END
```

§5. Тестовые примеры

$$CALL_\sqcup SFFRCS(C, S, 0.4)$$

Результат:

$$C(x) = 0.4966120676$$
$$S(x) = 0.06651848301$$

$$CALL_\sqcup SFFRCS(C, S, 13.0)$$

Результат:

$$C(x) = 0.5425104114$$
$$S(x) = 0.3982677211$$

Литература

1. Németh G. Chebyshev expansions for Fresnel integrals. Numer. Math., 1965, 7, № 4, 310–312.
2. Бахвалов Н.С. Об устойчивом вычислении значений многочленов. Ж. вычисл. матем. и матем. физ., 1971, 11, № 6, I568–1574.